
حعذد الوساساث 
 

 

هذا الكتٌب ٌتكلم عن تعدد المسارات فً الفٌجول باٌزٌك وٌفترض بالقارئ أنه ٌعرف 

 Classes & Interfacesأساسٌات اللغة والفبات والواجهات 

 

 

 :وٌحتوي على الموضوعات التالٌة

 Threading in Windows Forms Applications 

 المسارات بحٌرة استخدام Using the Thread pool 

 المسارات تزامن Thread Synchronization 

 النموذج على التحكمات فً النتٌجة وإظهار آخر مسار فً عملٌة تنفٌذ كٌفٌة 

  



Threading in Windows Forms Applications 

 
 

 هو أن التحكمات والنموذج ذات نفسه هو أنه ٌجب الوصول إلٌهم حصرٌا من Windows Forms تكمن المشكلة فً أغراض 

 وذلك بسبب أنها جمٌعا معتمدة على STA Model تعتمد على Windows Formsخلال المسار الذي قام بإنشابهم وفً الحقٌقة كل أغراض 

 مما ٌعنً أنه ٌمكنك إنشاء النموذج أو التحكم على أي مسار Apartment-Threaded والتً ترث مسارات الغرفة Win32هٌكلٌة رسابل 

مما ٌؤدي إلى ظهور العدٌد من المشاكل بسبب أن أقسام الدوت نٌت . ترٌده ولكن جمٌع الطرق المرتبطة به ٌجب استدعاؤها من نفس المسار
 ومزج كلا النوعٌن بدون حكمة تعتبر فكرة سٌبة وحتى لو لم تقم بإنشاء مسار بشكل واضح فً Free-Threading modelالأخرى تستخدم 

 من خلال UI Elementكودك ربما ستظهر لك بعض المشاكل فً جمٌع الأحوال فمثلا عندما تحاول الوصول إلى عنصر واجهة مستخدم 

  ٌتم تنفٌذها على مسار مختلف عن المسار الربٌسFinalizeً لنوع ما ونحن نعلم أن الطرٌقة Finalizeالطرٌقة 

 

The ISynchronizeInvoke Interface 

 ISynchronizeInvoke عناصر التحكم الوحٌدة التً ٌمكنك استدعاؤهم من مسار آخر هم الذٌن ٌتم عرضهم من خلال الواجهة 

حٌث تعٌد الخاصٌة .  القابلة للقراءة فقطInvokeRequired و الخاصٌة EndInvoke و BeginInvoke التً تمتلك الطرابق 

InvokeRequired القٌمة True إذا كان المستدعً لا ٌستطٌع الوصول إلى التحكم مباشرة وذلك عندما ٌعمل المستدعً على مسار مختلف 

 للوصول إلى أي عنصر خاص بالتحكم Invokeعن المسار الذي تم إنشاء التحكم فٌه ففً هذه الحالة ٌحب على المستدعً استدعاء الطرٌقة 

أو ٌمكن للمسار المستدعً استخدام الطرابق .  تنفٌذ الطرٌقةUIوهذه الطرٌقة متزامنة لهذا ٌتم إٌقاف المسار المستدعً حتى ٌكمل مسار 

BeginInvoke و EndInvokeلتنفٌذ العملٌة بشكل لا متزامن . 

 كمحدد ثانً إذا كانت Objectوٌمكنه أخذ مصفوفة من النوع  (Function أو Sub) إجراء مفوض ٌشٌر إلى طرٌقة Invokeتأخذ الطرٌقة 

 لهذا UIالطرٌقة تتوقع واحد أو أكثر من المحددات وتضمن هٌكلٌة نماذج وٌندوز أن الإجراء الذي ٌشٌر إلٌه المفوض ٌتم تنفٌذه فً المسار 

 . ٌمكنه بأمان الوصول إلى أي تحكم على النموذج
 حٌث ٌظهر لنا المثال التالً كٌف ٌمكننا زٌارة UI للوصول إلى تحكم من مسار غٌر المسار Invokeسنرى كٌف ٌمكننا استخدام الطرٌقة 

 وأول شا سنقوم بعمله هو تحدٌد طرٌقة Labelجمٌع المجلدات ضمن شجرة مجلد من مسار ثانوي بٌنما ٌتم إظهار السم المجلد فً تحكم 

 تقوم بعمل الإظهار المطلوب التً ٌمكنها أن تكون مجرد إجراء بسٌط
 

' This method must run in the main UI thread. 

Sub ShowMessage(ByVal msg As String) 

   Me.lblMessage.Text = msg 

   Me.Refresh() 

End Sub 

 
ثم نقوم بتحدٌد إجراء مفوض ٌشٌر لتلك الطرٌقة ومتغٌر ٌحمل كابن لذلك المفوض ٌكون معرفا على مستوى النموذج كً تتم مشاركته بٌن 

 جمٌع الطرابق ضمن النموذج
 

' A delegate that can point to the ShowMessage procedure 

Delegate Sub ShowMessageDelegate(ByVal msg As String) 

' An instance of the delegate 

Dim threadSafeDelegate As ShowMessageDelegate 

 
 Button لزر أوامر Clickوستحتاج لطرٌقة تبدأ المسار الثانوي مثلا إجراء معالجة الحدث 

 
' Parse the c:\Windows directory when the user clicks this button. 

Private Sub btnSearch_Click(ByVal sender As Object, ByVal e As EventArgs) _ 

      Handles btnSearch.Click 

   Dim t As New Thread(AddressOf SearchFiles) 

   t.Start("c:\windows") 

End Sub 

 
  lblMessageوأخٌرا تقوم بكتابة الكود الذي سٌعمل على المسار الثانوي حٌث أنه من الضروري لذلك الكود أن ٌستطٌع الوصول للتحكم 

 لأي تحكم Invoke أو الطرٌقة Form Class فً فبة النموذج Invoke وهذا ٌتم من خلال الطرٌقة ShowMessageباستدعاء الطرٌقة 

 موجود على النموذج والتً تكون مكافبة لها تماما
 

' (This method runs in a non-UI thread.) 



Sub SearchFiles(ByVal arg As Object) 

   ' Retrieve the argument. 

   Dim path As String = arg.ToString() 

   ' Prepare the delegate 

   threadSafeDelegate = New ShowMessageDelegate(AddressOf ShowMessage) 

   ' Invoke the worker procedure. (The result isn't used in this demo.) 

   Dim files As List(Of String) = GetFiles(path) 

   ' Show that execution has terminated. 

   Dim msg As String = String.Format("Found {0} files", files.Count) 

   Me.Invoke(threadSafeDelegate, msg) 

End Sub 

 

' A recursive function that retrieves all the files in a directory tree 

' (This method runs in a non-UI thread.) 

Function GetFiles(ByVal path As String) As List(Of String) 

   ' Display a message. 

   Dim msg As String = String.Format("Parsing directory {0}", path) 

   Me.Invoke(threadSafeDelegate, msg) 

   ' Read the files in this folder and all subfolders. 

   Dim files As New List(Of String) 

   For Each fi As String In Directory.GetFiles(path) 

      files.Add(fi) 

   Next 

   For Each di As String In Directory.GetDirectories(path) 

      files.AddRange(GetFiles(di)) 

   Next 

   Return files 

End Function 

 
 مثلا ٌمكن استدعاؤها من GetFiles على جمٌع المسارات فالطرٌقة ShowMessageوستتعقد العملٌة أكثر إن احتجنا لاستخدام الطرٌقة 

 ٌضٌف استباقا للأمور ٌجب تجنبه لذلك ٌجب علٌنا فحص قٌمة Invoke وفً هذه الحالة عمل الاستدعاء باستخدام الطرٌقة UIالمسار 

  False واستخدام الطرٌقة العادٌة إن كانت تعٌد القٌمة InvokeRequiredالخاصٌة 

 
' (Inside the SearchFiles and GetFiles methods) 

If Me.InvokeRequired Then 

   Me.Invoke(threadSafeDelegate, msg) 

Else 

   ShowMessage(msg) 

End If 

 
 من أجل كل مستدعً سنقوم بفحصها من داخل الطرٌقة  InvokeRequiredوالطرٌقة الأفضل من ذلك بدلا من فحص الخاصٌة 

ShowMessage 

 
' This method can run in the UI thread or in a non-UI thread. 

Sub ShowMessage(ByVal msg As String) 

   ' Use the Invoke method only if necessary. 

 

   If Me.InvokeRequired Then 

      Me.Invoke(threadSafeDelegate, msg) 

      Return 

   End If 

 

   Me.lblMessage.Text = msg 

   Me.Refresh() 

End Sub 

 
 بدون القلق ShowMessage ستحتاج فقط لاستدعاء lblMessageفبعد هذا التغٌٌر أي قطعة من الكود ستحتاج لإظهار رسالة على التحكم 

 حول أي مسار ٌتم تنفٌذ الكود علٌه
 ٌقوم التطبٌق بالوصول للتحكم عن طرٌق مسار غٌر مسار الإظهار 2 أو الفرٌموورك رقم 2005وفً بعض الظروف فً فٌجول باٌزٌك 

non-UI thread بدون التسبب بأٌة مشاكل فٌمكن حدوث ذلك مثلا عندما تحاول الوصول إلى تحكمات بسٌطة مثل Label أو عندما تقوم 



 فً الخلفٌة كما أن العدٌد من الخصابص ٌمكن قراءتها ولٌس تعدٌلها بدون التسبب بمشاكل وذلك Win32بعملٌات لا تسبب إرسال رسابل 

 لأن قٌمة تلك الخصابص مخزنة فً عنصر ضمن تحكم الدوت نٌت
 

The BackgroundWorker Component 

 تجنبك من الوقوع فً المشاكل المتعلقة بالمسارات فً تطبٌقات نماذج وٌندوز ISynchronizeInvoke على الرغم من أن الواجهة 

ٌحتاج معظم مطوري فٌجول باٌزٌك لطرٌقة أفضل وأقل أخطاء فأنت تحتاج مثلا لطرٌقة بسٌطة لإلغاء طرٌقة غٌر متزامنة بأسلوب آمن 

 إلى  BackgroundWorker ومن أجل هذا السبب قامت ماٌكروسوفت بإضافة المكون . الشا الذي لا توفره الواجهة المذكورة بشكل تلقابً

 . صندوق الأدوات واستخدامه سهل جدا مما ٌسهل عملٌة إنشاء تطبٌقات وٌندوز متعددة المسارات

 إذا أطلق True تكون قٌمتها WorkerReportsProgress خاصٌتان مثٌرتان للاهتمام فالخاصٌة BackgroundWorkerٌمتلك المكون 

 إذا كان المكون ٌدعم الطرٌقة True تكون قٌمتها WorkerSupportsCancellation والخاصٌة ProgressChangedالمكون الحدث 

CancelAsync وتكون القٌمة الافتراضٌة لكلا الطرٌقتٌن False لذا ٌجب علٌك ضبط قٌمتهم إلى True إذا أردت الاستفادة من جمٌع مزاٌا 

 BackgroundWorker وٌتطلب استخدام المكون Trueهذا التحكم والمثال الذي سٌطرح هنا ٌفترض أنه قد تم ضبط كلتا القٌمتٌن إلى 

 :بشكل عام العملٌات التالٌة

 وملؤها بالكود الذي ترٌد أن ٌتم تنفٌذه على المسار الثانوي وٌتم تشغٌل هذا الكود عندما ٌتم DoWorkإنشاء إجراء معالجة للحدث  .1

  حٌث لا ٌمكن للكود DoWork وهً تقبل محددا ٌتم تمرٌره لإجراء معالجة الحدث RunWorkerAsyncاستدعاء الطرٌقة 

 الموجود هناك الوصول مباشرة للتحكمات على النموذج لأنه ٌعمل فً مسار آخر

 عندما ترٌد الوصول إلى عنصر على النموذج وهذه الطرٌقة DoWork من داخل الحدث ReportProgressاستخدم الطرٌقة  .2

 وإلا سٌتم إطلاق استثناء True هً Worker-ReportsProgress إذا كانت قٌمة الخاصٌة ProgressChangedتطلق الحدث 

Worker-ReportsProgress فً حالة كون قٌمتها False والكود فً إجراء معالجة الحدث ProgressChanged ًٌعمل ف 

  ولهذا ٌمكنه الوصول بأمان لأي من تحكمات النموذجUIنفس المسار 

 لإٌقاف المسار الثانوي مباشرة وهذه الطرٌقة تستدعً ضبط BackgroundWorker للتحكم CancelAsyncاستخدم الطرٌقة  .3

 فً حالة InvalidOperationException وإلا سٌتم إطلاق استثناء True إلى WorkerSupportsCancellationالخاصٌة 

 والخروج بأمان عندما CancellationPending التحقق دورٌا من الخاصٌة DoWork وٌجب على الكود فً Falseكون قٌمتها 

 Trueتصبح قٌمتها 

 إن كنت ترٌد القٌام بأٌة أعمال عندما ٌنتهً عمل المسار الثانوي إما RunWorkerCompletedكتابة إجراء معالجة للحدث  .4

 لذا ٌستطٌع الوصول لجمٌع عناصر UIبشكل طبٌعً أو بواسطة الإلغاء والكود فً إجراء معالجة هذا الحدث ٌعمل فً المسار 

 النموذج

 ٌجب أن ٌعٌد قٌمة للمسار الأساسً بدلا من تعٌٌن هذه القٌمة فً حقل على مستوى الفبة فعلى DoWorkوبشكل عام فالكود فً معالج الحدث 

 فتكون هذه القٌمة متوفرة للمسار الأساسً بواسطة الخاصٌة DoWorkEventArgs للغرض Resultالكود تعٌٌن هذه القٌمة للخاصٌة 

Result للغرض RunWorkerCompletedEventArgs الممرر للحدث RunWorkerCompleted وهذا كود نموذجً ٌستخدم 

 BackgroundWorkerالعنصر 

 

' The button that starts the asynchronous operation 

Private Sub btnStart_Click(ByVal sender As Object, ByVal e As EventArgs) _ 

      Handles btnStart.Click 

   Dim argument As Object = "abcde"           ' The argument 

 

   BackgroundWorker1.RunWorkerAsync(argument) 

   ' Disable this button, and enable the "Stop" button. 

   btnStart.Enabled = False 

   btnStop.Enabled = True 

End Sub 

 

' The button that cancels the asynchronous operation 

Private Sub btnStop_Click(ByVal sender As Object, ByVal e As EventArgs) _ 

      Handles btnStop.Click 

   BackgroundWorker1.RunWorkerAsync(argument) 

End Sub 

 



' The code that performs the asynchronous operation 

Private Sub BackgroundWorker1_DoWork(ByVal sender As Object, _ 

      ByVal e As DoWorkEventArgs) Handles BackgroundWorker1.DoWork 

   ' Retrieve the argument. 

   Dim argument As Object = e.Argument 

   Dim percentage As Integer = 0 

   … 

   ' The core of the asynchronous task 

   Do Until BackgroundWorker1.CancellationPending 

      … 

      ' Report progress when it makes sense to do so. 

      BackgroundWorker1.ReportProgress(percentage) 

   Loop 

   ' Return the result to the caller. 

   e.Result = primes 

End Sub 

 

' This method runs when the ReportProgress method is invoked. 

Private Sub BackgroundWorker1_ProgressChanged(ByVal sender As Object, _ 

             ByVal e As ProgressChangedEventArgs) Handles _ 

             BackgroundWorker1.ProgressChanged 

 

   ' It is safe to access the user interface from here. 

   ' For example, show the progress on a progress bar or another control. 

   ToolStripProgressBar1.Value = e.ProgressPercentage 

End Sub 

 

' This method runs when the asynchronous task is completed (or canceled). 

Private Sub BackgroundWorker1_RunWorkerCompleted(ByVal sender As Object, _ 

                ByVal e As RunWorkerCompletedEventArgs) Handles _ 

                BackgroundWorker1.RunWorkerCompleted 

   ' It is safe to access the user interface from here. 

   … 

   ' Reset the Enabled state of the Start and Stop buttons. 

   btnStart.Enabled = True 

   btnStop.Enabled = False 

End Sub 

 
 للبحث عن الملفات فً مسار غٌر متزامن وهً نفس المشكلة BackgroundWorkerٌظهر لك المثال التالً كٌف ٌمكن استخدام العنصر 

.  فً هذا الموضوع سابقا وبهذا ٌمكنك مقارنة الطرٌقتٌن بسهولةThe ISynchronizeInvoke Interfaceالتً طرحت عند الحدٌث عن 

  أكثر تعقٌدا بقلٌل بسبب أنها تدعم الإلغاء لعمل غٌر متزامنBackgroundWorkerوستكون النسخة الجدٌدة المعتمدة على 

' The result from the SearchFiles procedure 

Dim files As List(Of String) 

' We need this variable to avoid nested calls to ProgressChanged. 

Dim callInProgress As Boolean 

 

' The same button works as a Start and a Stop button. 

Private Sub btnStart_Click(ByVal sender As Object, ByVal e As EventArgs) _ 

      Handles btnStart.Click 

   If btnStart.Text = "Start" Then 

      lstFiles.Items.Clear() 

      Me.BackgroundWorker1.RunWorkerAsync("c:\windows") 

      Me.btnStart.Text = "Stop" 

   Else 

      Me.BackgroundWorker1.CancelAsync() 

   End If 

End Sub 

 

' The code that starts the asynchronous file search 

Private Sub BackgroundWorker1_DoWork(ByVal sender As Object, _ 

          ByVal e As DoWorkEventArgs) Handles BackgroundWorker1.DoWork 

 



   ' Retrieve the argument. 

   Dim path As String = e.Argument.ToString() 

   ' Invoke the worker procedure. 

   files = New List(Of String) 

   SearchFiles(path) 

   ' Return a result to the RunWorkerCompleted event. 

   Dim msg As String = String.Format("Found {0} files", files.Count) 

   e.Result = msg 

End Sub 

 

' A recursive function that retrieves all the files in a directory tree. 

Sub SearchFiles(ByVal path As String) 

   ' Display a message. 

 

   Dim msg As String = String.Format("Parsing directory {0}", path) 

   ' Notice that we don't really use the percentage; 

   ' instead, we pass the message in the UserState property. 

   Me.BackgroundWorker1.ReportProgress(0, msg) 

 

   ' Read the files in this folder and all subfolders. 

   ' Exit immediately if the task has been canceled. 

   For Each fi As String In Directory.GetFiles(path) 

      If Me.BackgroundWorker1.CancellationPending Then Return 

      files.Add(fi) 

   Next 

   For Each di As String In Directory.GetDirectories(path) 

       If Me.BackgroundWorker1.CancellationPending Then Return 

       SearchFiles(di) 

   Next 

End Sub 

 

 

Private Sub BackgroundWorker1_ProgressChanged(ByVal sender As Object, _ 

      ByVal e As ProgressChangedEventArgs) _ 

      Handles BackgroundWorker1.ProgressChanged 

   ' Reject nested calls. 

   If callInProgress Then Return 

   callInProgress = True 

   ' Display the message, received in the UserState property. 

   Me.lblMessage.Text = e.UserState.ToString() 

   ' Display all files added since last call. 

   For i As Integer = lstFiles.Items.Count To files.Count - 1 

      lstFiles.Items.Add(files(i)) 

   Next 

   Me.Refresh() 

   ' Let the Windows operating system process message in the queue. 

   ' If you omit this call, clicks on buttons are ignored. 

   Application.DoEvents() 

   callInProgress = False 

End Sub 

 

Private Sub BackgroundWorker1_RunWorkerCompleted(ByVal sender As Object, _ 

      ByVal e As RunWorkerCompletedEventArgs) _ 

      Handles BackgroundWorker1.RunWorkerCompleted 

   ' Display the last message and reset button's caption. 

   Me.lblMessage.Text = e.Result.ToString() 

   btnStart.Text = "Start" 

End Sub 

 
 حٌث ٌجب أن ٌتضمن الكود استدعاء للطرٌقة ProgressChangedوالكود هنا ٌشرح نفسه ماعدا إجراء الحدث 

Application.DoEvents وإلا لن ٌتمكن التطبٌق من معالجة الأحداث المنطلقة مثل حدث النقر على الزر Stop أو أي عمل آخر ممكن 

 مما قد ٌسبب إطلاق استثناء ProgressChangedإضافته للواجهة ومع ذلك فاستدعاء هذه الطرٌقة سٌسبب استدعاءات معششة للإجراء 



StackOverflowException ومن أجل عدم حدوث هذا ٌتم استخدام حقل منطقً مساعد callInProgress لتجنب حدوث مثل هذه 

 الاستدعاءات المعششة 
 فقط لتنفٌذ جزء من ReportProgressلاحظ أٌضا أن هذا التطبٌق لا ٌحتاج للإعلام عن نسبة التقدم للمسار الربٌسً وٌستخدم الطرٌقة 

 أو progress bar وإن كان تطبٌقك ٌستخدم UserStateالكود فً المسار الربٌسً للبرنامج والرسالة الفعلٌة للإظهار ٌتم تمرٌرها للخاصٌة 

 بدون داعً لأنها تتسبب بتبدٌل المسارات وتكون مكلفة كثٌرا ReportProgressأي مؤشر آخر للتقدم ٌجب علٌك تجنب استدعاء الطرٌقة 

عندما ٌتعلق الأمر بوقت المعالجة وفً هذه الحالة ٌجب علٌك تخزٌن مؤشر التقدم فً حقل فً الفبة واستدعاء الطرٌقة فقط فً حالة حدوث 
 تقدم فعلً

 
Dim currentPercentage As Integer 

 

Private Sub BackgroundWorker1_DoWork(ByVal sender As Object, _ 

             ByVal e As DoWorkEventArgs) Handles BackgroundWorker1.DoWork 

 

   Const TotalSteps = 5000 

   For i As Integer = 1 To TotalSteps 

      … 

      ' Evaluate progress percentage. 

      Dim percentage As Integer = (i * 100) \ TotalSteps 

      ' Report to UI thread only if percentage has changed. 

      If percentage <> currentPercentage Then 

         BackgroundWorker1.ReportProgress(percentage) 

         currentPercentage = percentage 

      End If 

   Next 

End Sub 

 
 

  



 Using the Thread poolاسخخذام بحيشة الوساساث 

 
 

 إنشاء العدٌد من المسارات قد ٌسبب انخفاض أداء النظام بسرعة وخاصة عندما تصرف هذه المسارات معظم وقتها فً حالة سبات 

ولتحسٌن أداء كودك ٌمكنك إعادة ترتٌب بحٌرة المسارات بشكل . أوٌعاد تشغٌلها بصورة دورٌة بغرض قراءة مصدر ما أو تحدٌث الإظهار

  بحٌرة المساراتSystem.Threading الموجودة فً مجال الأسماء Objectsٌضمن أكفأ استخدام للموارد باستخدام بعض الأغراض 

 

The ThreadPool Type 

 والتً تحتاج لإجراء مفوض ThreadPool.QueueUserWorkItem ٌتم إنشاء بحٌرة المسارات عندما تقوم باستدعاء الدالة 

WaitCallback delegate وغرض Object اختٌاري ٌستخدم لتمرٌر البٌانات للمسار والإجراء المفوض ٌجب أن ٌشٌر إلى Sub ٌمرر 

 عندما لا توجد بٌانات Nothing بحٌث تكون قٌمته محتوٌة على البٌانات التً نرٌد تمرٌرها للمسار أو Objectله محدد وحٌد من النوع 

 Classنرٌد تمرٌرها وقطعة الكود التالٌة تبٌن لك كٌف ٌمكنك استخدام عدد كبٌر من المسارات لاستدعاء إجراء فً فبة 

 
For i As Integer = 1 To 20 

     ' Create a new object for the next lightweight task. 

      Dim task As New LightweightTask() 

      ' Pass additional information to it. (Not used in this demo.) 

       task.SomeData = "other data" 

       ' Run the task with a thread from the pool.  

       ' (Pass the counter as an argument.) 

       ThreadPool.QueueUserWorkItem(AddressOf task.Execute, i) 

 Next 

 
 وقطعة الكود التالٌة تحتوي على الكود الذي ٌتم تنفٌذه فعلا عندما ٌتم سحب المسار من البركة

 
Class LightweightTask 

    Public SomeData As String 

 

    ' The method that contains the interesting code 

    ' (Not really interesting in this example) 

    Sub Execute(ByVal state As Object) 

        Console.WriteLine("Message from thread #{0}", state) 

    End Sub 

End Class 

 
والمسار العامل ٌمكنه تحدٌد فٌما إذا كان قد أخذ من بحٌرة المسارات أم لا بتحري قٌمة الخاصٌة 

Thread.CurrentThread.IsThreadPoolThread وٌمكنك معرفة العدد الأقصى للمسارات فً البركة باستدعاء الطرٌقة الساكنة 

ThreadPool.GetMaxThreads وعدد المسارات المتاحة حالٌا باستدعاء الطرٌقة الساكنة ThreadPool.GetAvailableThreads . 

 تمكنك من تغٌٌر العدد الأقصى للمسارات NET Framework 2.0. 2 فً الفرٌموورك SetMaxThreadsكما تم إضافة طرٌقة جدٌدة 

 الموجودة فً البركة 

 
' Maximum 30 worker threads and maximum 10 asynchronous I/O threads in the pool 

ThreadPool.SetMaxThreads(30, 10) 

 
استخدم : وتظهر هنا قاعدة جٌدة. فً بعض الأحٌان قد تحتار فً نقطة تساؤل هل أقوم بإنشاء المسار بنفسً أم أستعٌره من بحٌرة المسارات

 عندما ترٌد تنفٌذ عملٌة ترٌد تنفٌذها بأسرع وقت أو عندما ترٌد القٌام بعملٌة تستهلك الوقت ولا ٌتم تنفٌذها Thread classفبة المسارات 

 .كثٌرا وفً معظم الحالات بشكل عام ٌجب علٌك استخدام بحٌرة المسارات



 

The Timer Type 

فمثلا ٌجب علٌك استخدام التحكم .  تقدم الفرٌموورك عدة أنواع من المؤقتات كل منها ٌمتلك نقاط قوته وضعفه

System.Windows.Forms.Timer عندما تقوم بالعمل على تطبٌق من النوع Windows Forms applications وإن لم ٌكن برنامجك 

 وتعتبر هاتان System.Timers.Timer أو الفبة System.Threading.Timerٌمتلك واجهة للمستخدم ٌجب علٌك عندها استخدام الفبة 

 System.Timers.Timer ٌنطبق أٌضا على System.Threading.Timerالفبتان متساوٌتٌن فً العمل تقرٌبا والشرح التالً على 

 
 ٌقدم طرٌقة بسٌطة لمؤقت ٌستدعً إجرابٌة محددة حٌث ٌمكنك استخدام System.Threading فً مجال الأسماء Timer الفبة 

هذه الفبة لجدولة عمل فً وقت معٌن فً المستقبل وٌمكن تنفٌذه بالتكرار الذي تحتاجه مهما ٌكن ابتداء من مرة واحدة فما فوق وبانً المؤقت 

 :ٌأخذ أربعة محددات

  إجراء مفوضTimerCallback delegate ٌشٌر إلى الإجراء الذي ٌستدعى عندما ٌنتهً زمن المؤقت وٌجب أن ٌكون هذا 

 Object ٌأخذ محدد واحد من النوع Subالإجراء من النوع 

  غرضObject ٌتم تمرٌره للإجراء الذي ٌشٌر  إلٌه المفوض وٌمكن أن ٌكون من عدة أنواع كسلسلة نصٌة أو مصفوفة أو 

 أو أي نوع بٌانات آخر ٌحتوي على البٌانات التً سٌتم تمرٌرها للإجراء وإن لم تكن تحتاج لتمرٌر قٌم Collectionمجموعة 

  بكل بساطةNothingاستخدم 

  قٌمة من النوعTimeSpan تحدد زمن المؤقت الذي سٌتم استدعاء الإجراء بعده كما ٌمكن تحدٌدها باستخدام قٌمة من النوع Long 

 كقٌمة لا ٌتم Timeout.Infiniteوعند تمرٌر  ( من الثانٌة1/1000) وفً هذه الحالة ٌقاس الزمن بالمٌللً ثانٌة UInteger أو 

  صفر لإطلاق المؤقت مباشرة0إطلاق المؤقت أبدا أو القٌمة 

  قٌمة من النوعTimeSpanوهذه أٌضا ٌمكن .  تحدد زمن المؤقت والتً بدورها تحدد زمن تكرار إطلاق المؤقت بعد المرة الأولى

 أو 1 وهنا أٌضا ٌصبح الوقت مقاسا بالمٌللً ثانٌة وٌمكنك تمرٌر القٌمة UInteger أو Longتحدٌدها بقٌمة من النوع 

Timeout.Infiniteلإطلاق المؤقت مرة واحدة فقط . 

 Changeوبعد تشغٌل المؤقت ٌمكنك تغٌٌر هذه القٌم فقط باستخدام الطرٌقة . وهذه القٌم التً تمررها لبانً المؤقت  غٌر متوفرة كخصابص

method  والتً تأخذ محددٌن ٌحددان وقت التشغٌل وفترة زمن المؤقت وٌمتلك Timer object إجراء Stop الذي ٌقوم بإٌقاف المؤقت 

  وترٌنا قطعة الكود التالٌة مثالا عما تحدثنا عنه حول المؤقتDisposeالذي ٌتم إٌقافه عبر استدعاء الإجراء 

 
Sub TestThreadingTimer() 

    ' Get the first callback after one second. 

    Dim dueTime As New TimeSpan(0, 0, 1) 

    ' Get additional callbacks every half second. 

    Dim period As New TimeSpan(0, 0, 0, 0, 500) 

    ' Create the timer. 

    Using t As New Timer(AddressOf TimerProc, Nothing, dueTime, period) 

        ' Wait for five seconds in this demo, and then destroy the timer. 

        Thread.Sleep(5000) 

    End Using 

End Sub 

 

' The callback procedure 

Sub TimerProc(ByVal state As Object) 

    ' Display current system time in console window. 

    Console.WriteLine("Callback proc called at {0}", Date.Now) 

End Sub 

 
وفً النهاٌة تجدر ملاحظة أن الإجراء المستدعى ٌتم تنفٌذه على مسار مأخوذ من بركة المسارات لذا ٌجب علٌك التحكم بالمتغٌرات 

 والمصادر الأخرى المستخدمة من قبل المسار الربٌسً للبرنامج عبر استخدام ما ٌدعى بتزامن المسارات

  



 Thread Synchronizationحزاهن الوساساث 
 

The SyncLock Statement 

خلال زمن التشغٌل لا ٌوجد شا ٌضمن لك أن ٌسٌر الكود بشكل نظامً بدون مقاطعات وتكون عملٌة التشغٌل بدون مقاطعات عملٌة قاسٌة 

على نظام التشغٌل وخاصة عندما ٌكون عبارة عن بٌبة متعددة المهام وفً معظم الحالات التً ستحتاجها ستكون قانعا بالدقة ضمن البرنامج 

الواحد وذلك عند معالجة الكود فعلى سبٌل المثال ٌكون كافٌا لك ضمان أن مسار تنفٌذ واحد ضمن التطبٌق الحالً ٌستطٌع تنفٌذ قطعة معٌنة 

 والذي ٌحتاج إلى متغٌر SyncLock…End SyncLockمن الكود فً وقت محدد وٌمكنك تحقٌق ذلك بتضمٌن قطعة الكود تلك ضمن كتلة 

 :كمحدد له محققا المتطلبات التالٌة

  ٌجب أن ٌكون مشترك بٌن جمٌع المسارات وٌكون فً العادة متغٌر على مستوى الفبة وبدون الخاصٌةThreadStatic 

  ٌجب أن ٌكون من نوع مرجعً مثلString أو Objectواستخدام أنواع القٌمة ٌنتج عنه خطأ فً الترجمة  

  ٌجب أن لا ٌحتوي على القٌمةNothing وفً حال تمرٌر القٌمة Nothingسٌسبب أخطاء فً زمن التنفٌذ  

 SyncLock وفٌما ٌلً مثال عن كتلة 

 
' The lock object. (Any non-Nothing reference value will do.) 

Private consoleLock As New Object() 

 

Sub SynchronizationProblem_Task(ByVal obj As Object) 

   Dim number As Integer = CInt(obj) 

   ' Print a lot of information to the console window. 

   For i As Integer = 1 To 1000 

      SyncLock consoleLock 

         ' Split the output line in two pieces. 

         Console.Write(" ") 

         Console.Write(number) 

      End SyncLock 

   Next 

End Sub 

 
 وهو ٌشكل المصدر الوحٌد المشترك بٌن جمٌع Console للتحكم بالوصول للغرض consoleLockوالكود السابق ٌستخدم المتغٌر 

المسارات فً المثال ولهذا فهو المصدر الذي ٌجب علٌك تحقٌق التزامن من أجله والتطبٌقات الحقٌقٌة ٌمكن أن تحوي العدٌد من كتل 

SyncLock والتً ٌمكن أن تستخدم نفس المتغٌر المحلً أو عدة متغٌرات مختلفة من أجل اختلاف البصمة وهنا ٌجب علٌك استخدام متغٌرا 

ممٌزا من أجل كل نوع من أنواع المصادر المشتركة التً ٌجب عمل التزامن من اجلها أو من أجل مجموعة التعابٌر التً ٌجب تنفٌذها ضمن 

 . المسار فً نفس الوقت

 مخفٌة من أجل ضمان تحرٌر القفل بشكل صحٌح إذا تم إطلاق Try…End Try ٌتضمن الكود  تلقابٌا كتلة SyncLockوعندما تستخدم كتلة 

 موضوعة داخل إجراء خاص بتواجد SyncLockوإن كانت الكتلة . SyncLockاستثناء ومن أجل هذا لا ٌمكنك القفز لعبارة داخل الكتلة 

Instance لفبة ما وجمٌع المسارات العاملة ضمن إجراء فً ذلك التواجد Instance للفبة ٌمكنك تمرٌر Me لعبارة الـ SyncLock وذلك 

 (Nothing وبالتأكٌد هو لٌس – وهو قٌمة مرجعٌة –ٌمكن الوصول إلٌه من جمٌع المسارات )بسبب أن هذا الغرض ٌحقق كل المتطلبات 

 
Class TestClass 

   Sub TheTask() 

      SyncLock Me 

         ' Only one thread at a time can access this code. 

         … 

      End SyncLock 

   End Sub 

End Class 

 
 Console بهذه الطرٌقة فقط إن كنت ترٌد عمل التزامن على مصدر وحٌد كملف محدد مثلا أو نافذة الكونسول Meٌمكنك استخدام : ملاحظة

Window وإن كان لدٌك عدة كتل تزامن التً تحمً عدة مصادر ستستخدم بشكل تلقابً عدة متغٌرات كمحددات لكتلة SyncLock . والشا



 كمحدد فقط إذا كانت الفبة غٌر مربٌة خارج المجمع الحالً عدا ذلك ٌمكن Meالذي له أهمٌة أكبر مما ذكر هو أنه ٌجب علٌك استخدام 

 مختلفة وبهذا فلن ٌتم تنفٌذ عدة كتل من الكود بدون سبب حقٌقً SyncLock للفبة ضمن كتلة Instanceلتطبٌق آخر استخدام نفس التواجد 

وتجدر الملاحظة أن . SyncLock عام مربً من مجمعات أخرى كمحدد لكتلة Objectمحدد وبشكل عام لا ٌجب علٌك استخدام غرض 

 وذلك lock object للحصول على نوع الغرض المستخدم للقفل GetTypeالعدٌد من الأكواد التً تراها على الانترنت تستخدم العامل 

 .لحماٌة الطرٌقة الساكنة

 معششة للقٌام بالتزامن لأغراض مختلفة من الضروري استخدام تسلسل تعشٌش متطابق أٌنما احتجت له SyncLockعندما تستخدم عبارات 

فً تطبٌقك فالتحري عن الأقفال بالتسلسل المطابق ذاته ٌجنبك الوصول إلى حالة الأقفال المٌتة خلال العدٌد من أجزاء التطبٌق وهذه القاعدة 

 SyncLock باستدعاء دالة أخرى تحتوي على SyncLockتنطبق أٌضا عندما تقوم دالة تحتوي على 

 
' Always use this sequence when locking objLock1 and objLock2. 

SyncLock objLock1 

   SyncLock objLock2 

      … 

   End SyncLock 

End SyncLock 

 

 Performance Considerations and Lazy Instantiationاعخباساث الأداء والخىاجذ الكسىل 

 ٌؤدي إلى إبطاء تطبٌقك كثٌرا أو تخفٌض أداؤه بشكل SyncLock تضمٌن جمٌع الأكواد التً تستخدم متغٌرات مشتركة ضمن كتلة 

 بدون تعرٌض تكامل SyncLockملحوظ وبشكل خاص عندما ٌتم تشغٌله على حاسب متعدد المعالجات فإن استطعت تجنب استخدام كتلة 

  فً بٌبة متعددة المساراتlazy instantiationالبٌانات للخطر ٌجب علٌك القٌام به قطعٌا فمثلا تخٌل أنك تستخدم نمط وحٌد بتواجد كسول 

 
Public Class Singleton 

   Private Shared m_Instance As Singleton 

   Private Shared sharedLock As New Object() 

 

   Public Shared ReadOnly Property Instance() As Singleton 

      Get 

         SyncLock sharedLock 

            If m_Instance Is Nothing Then m_Instance = New Singleton 

            Return m_Instance 

         End SyncLock 

      End Get 

   End Property 

End Class 

 
 ٌتم تعٌٌنه مرة  m_Instanceتكمن المشكلة فً الكود السابق أن معظم الوصولات للخاصٌة لا ٌحتاج إلى تزامن وذلك لأن المتغٌر الخاص 

 واحدة فً المرة الأولى التً ٌتم فٌها قراءة الخاصٌة وفً ما ٌلً طرٌقة أفضل لتحقٌق التصرف المطلوب

 
Class Singleton 

   Private Shared m_Instance As Singleton 

   Private Shared sharedLock As New Object 

 

   Public Shared ReadOnly Property Instance() As Singleton 

      Get 

         If m_Instance Is Nothing Then 

            SyncLock sharedLock 

               If m_Instance Is Nothing Then m_Instance = New Singleton() 

            End SyncLock 

         End If 

         Return m_Instance 

      End Get 

   End Property 

End Class 



 

 Synchronized Objectsالأغشاض الوخزاهنت 

 قابلة للمشاركة بأمان عبر NET object. مشكلة أخرى متعلقة بالمسارات فً الدوت نٌت هً أن لٌس جمٌع أغراض الدوت نٌت 

فعندما تقوم بكتابة تطبٌق متعدد المسارات ٌجب علٌك التأكد دوما من الوثابق للتأكد not all .NET objects are thread-safe المسارات 

 Match و Regexمن أن الأغراض والطرابق التً تستخدمها آمنة للاستخدام عبر المسارات فعلى سبٌل المثال جمٌع الطرق الساكنة للفبات 

أمنة عبر المسارات ولكن الطرق الغٌر ساكنة غٌر آمنة فٌجب عدم استخدامها ضمن مسار مختلف وكذلك بعض أغراض الدوت نٌت مثل 

Windows Forms objects and controlsلها العدٌد من الحدود التً تجعل فقط المسار الذي أنشأها ٌمكنه استدعاء طرقها وخصابصها  

 

 Synchronized .NET Typesأنىاع دوث نيج الوخزاهنت 

  و Stack و SortedList و Queue و Hashtable و ArrayList  العدٌد من الأغراض الغٌر آمنة عبر المسارات بطبٌعتها مثل 

TextReader و TextWriter التعابٌر النظامٌة تقدم طرٌقة ساكنة قابلة للتزامن تعٌد غرض أمن للمسارات  وthread-safe object 

  عندما تتعامل مع نسخة آمنة عبر المساراتTrue التً تعٌد IsSynchronizedمكافا للذي تم تمرٌره كما أن معظمها ٌعرض الخاصٌة 

 
' Create an ArrayList object, and add some values to it. 

Dim al As New ArrayList() 

al.Add(1): al.Add(2): al.Add(3) 

' Create a synchronized, thread-safe version of this ArrayList. 

Dim syncAl As ArrayList = ArrayList.Synchronized(al) 

' Prove that the new object is thread-safe. 

Console.WriteLine(al.IsSynchronized)        ' => False 

Console.WriteLine(syncAl.IsSynchronized)    ' => True 

' You can now share the syncAl object among different threads 

 
تذكر دابما أن التعامل مع هذه النسخة المتزامنة ٌكون أبطأ من النسخة الغٌر متزامنة وذلك بسبب أن كل طرٌقة تمر عبر سلسلة من 

 regular arrays andالفحوصات الداخلٌة وفً معظم الحالات ٌمكنك كتابة كود فعال أكثر إذا استخدمت المصفوفات والمجموعات العادٌة 

collections وقمت بمزامنة عناصرها باستخدام كتلة  SyncLockالعادٌة  

 

The Synchronization Attribute 

 هً أبسط طرٌقة لتحقٌق الوصول System.Runtime.Remoting.Contexts.Synchronization استخدام الخاصٌة 

 بأكمله وبذلك ٌستطٌع مسار واحد فقط الوصول إلى حقوله وطرابقه وبذلك أي مسار ٌستطٌع استخدام الفبة ولكن Objectالمتزامن للغرض 

 وأي مسار ٌحاول استخدام هذه الفبة علٌه الانتظار Classمسار واحد فقط ٌستطٌع تنفٌذ أحد طرابقه إذا كانت الطرٌقة تنفذ كودا ضمن الفبة 

والكود التالً ٌبٌن كٌف ٌمكنك مزامنة فبة .  تغلف كافة طرابق الفبة مستخدمة نفس متغٌر الإقفالSyncLockوبكلمات أخرى وكأن هناك كتل 

 لٌتم تعلٌمها كـ ContextBoundObject لاحظ أٌضا أن الفبة ٌجب أن ٌتم وراثتها من Synchronization attributeباستخدام الخاصٌة 

context-bound object 

 
System.Runtime.Remoting.Contexts.Synchronization()> _ 

Class Display 

   Inherits ContextBoundObject 

   … 

End Class 

 
 تضمن الوصول المتزامن لجمٌع الحقول والخصابص والطرق ولكنها لا توفر التزامن Synchronization attributeو خاصٌة التزامن 

 أو أحد الثوابت التً توفرها الفبة False أو True وهً تأخذ محددا اختٌارٌا ٌمكن أن تكون قٌمته static membersللأعضاء الساكنٌن 

SynchronizationAttribute والتً ٌمكنك الاطلاع علٌها من مكتبة MSDN 

 



The MethodImpl Attribute 

 فً معظم الحالات مزامنة فبة كاملة ستقتل التطبٌق وحماٌة بعض الطرابق فً تلك الفبة ٌكون كافٌا فً معظم الحالات حٌث ٌمكنك 

 أو ٌمكنك استخدام تقنٌة أبسط مبنٌة على الصفة SyncLockتطبٌق هذا بتغلٌف كود الطرٌقة بكتلة 

System.Runtime.CompilerServices.MethodImpl 

 
Class MethodImplDemoClass 

    ' This method can be executed by one thread at a time. 

    <MethodImpl(MethodImplOptions.Synchronized)> _ 

    Sub SynchronizedMethod() 

      … 

    End Sub 

End Class 

 
 والتً تستخدم SyncLock على عدة طرابق فً الفبة ٌؤدي نفس الغرض من تغلٌف كامل تلك الطرابق بكتلة MethodImplفتطبٌق الصفة 

Me كمتغٌر إقفال وبكلمات أخرى أي مسار ٌستدعً طرٌقة معلمة بالخاصٌة MethodImpl سوف ٌمنع أي مسار آخر من استدعاء 

 كما ٌمكنك استخدام هذه الخاصٌة على الطرابق الساكنة وٌكون متغٌر الغرض الذي ٌستخدم MethodImplالطرٌقة المعلمة بالخاصٌة 

 وبهذا فالمسار الذي instance methodsضمنٌا لقفل الطرابق الساكنة مختلف عن متغٌر الغرض المستخدم لقفل الطرابق الأخرى للفبة 

 والمعلمة instance methods لا ٌمنع مسار آخر من استدعاء الطرابق الغٌر ساكنة MethodImplٌستدعً طرٌقة ساكنة معلمة بالصفة 

 بنفس الصفة

 

 Volatile Read and Write Operationsعولياث القشاءة والكخابت الوخغيشة 

 عندما تتم مشاركة متغٌر عبر عدة مسارات والتطبٌق ٌعمل على حاسب متعدد المعالجات ٌجب علٌك وضع احتمال حدوث أخطاء 

إضافٌة فً الحسبان وتكمن المشكلة فً النظام متعدد المعالجات فً أن لكل معالج الكاش الخاص به ولهذا فإذا قمت بالكتابة على حقل فً فبة 

على مسار سٌتم كتابة القٌمة الجدٌدة فً الكاش المرتبط مع المعالج الحالً ولا ٌتم نشرها مباشرة إلى الكاش الخاص ببقٌة المعالجات بحٌث 

 بت الذي ٌمكنه إعادة ترتٌب تنفٌذ كتل عبارات 64كما تحدث مشكلة مشابهة فً الأنظمة ذات المعالج . ٌمكنهم جمٌعا رؤٌة القٌمة الجدٌدة

الكود متضمنا عملٌات القراءة والكتابة فً الذاكرة وعملٌة إعادة الترتٌب لم ٌكن لها تأثٌر ظاهر حتى الآن من أجل مسار واحد ٌستخدم جزءا 

وتوفر الفرٌموورك . معٌنا من الذاكرة ولكن ربما سٌسبب ذلك مشكلة عندما ٌتم الوصول إلى نفس الجزء من الذاكرة بواسطة عدة مسارات

 وٌوفرها جمٌعا النوع MemoryBarrier والطرٌقة VolatileWrite و VolatileReadحلان لهذه المشكلة وهما وزج من الطرابق 

Thread 

 من كتابة متغٌر والتأكد من أن القٌمة الجدٌدة ٌتم كتابتها آلٌا فً الذاكرة المشتركة بٌن جمٌع المعالجات ولا VolatileWriteتمكنك الطرٌقة 

 من قراءة المتغٌر بطرٌقة VolatileReadتبقى فً المسجل الخاص بالمعالج حٌث تكون مخفٌة عن بقٌة المسارات وبالمثل تمكنك الطرٌقة 

 بحٌث Overloadedآمنة لأنها تجبر النظام على تفرٌغ جمٌع ذواكر الكاش الموجودة قبل تنفٌذ العملٌة وكلا الطرٌقتان محملتان تحمٌلا زابدا 

  وبالمرجع كما فً قطعة الكود التالٌةObjectتأخذ متغٌرات رقمٌة أو غرضٌة 

 
Class TestClass 

    Private Shared sharedValue As Integer 

 

    Function IncrementValue() As Integer 

        Dim value As Integer = Thread.VolatileRead(sharedValue) 

        value += 1 

        Thread.VolatileWrite(sharedValue, value) 

        Return value 

    End Function 

End Class 

 
 ولكن لا ٌمكن استخدامهما من أجل Objectوالطرٌقتان المذكورتان تعملان بشكل جٌد عندما نتعامل مع المتغٌرات الرقمٌة أو الغرضٌة 

 بسبب عدم إمكانٌة الاعتماد على عملٌة Objectأنواع أخرى من المتغٌرات لأنه لا ٌمكنك استخدام نسخة الدالة التً تأخذ متغٌر من النوع 

 التً تقوم بتفرٌغ محتوٌات جمٌع ذواكر الكاش MemoryBarrierالتحوٌل عندما ٌكون المتغٌر ممررا بالمرجع مما ٌقودنا إلى الطرٌقة 



الخاصة بالمعالجات إلى الذاكرة الربٌسٌة وبهذا تضمن لك أن جمٌع المتغٌرات تحتوي أحدث نسخة من البٌانات التً تمت كتابتها إلٌهم فمثلا 

  تعمل جٌدا حتى على نظام متعدد المعالجاتSingletonٌضمن الكود التالً أن الفبة 

 
Class Singleton 

    Private Shared m_Instance As Singleton 

    Private Shared sharedLock As New Object() 

 

    Public Shared ReadOnly Property Instance() As Singleton 

        Get 

            If m_Instance Is Nothing Then 

                SyncLock sharedLock 

                    If m_Instance Is Nothing Then 

                        Dim tempInstance As Singleton = New Singleton() 

                      ' Ensure that writes related to instantiation are flushed. 

                        Thread.MemoryBarrier() 

                        m_Instance = tempInstance 

                    End If 

                End SyncLock 

            End If 

            Return m_Instance 

        End Get 

    End Property 

End Class 

 

 مباشرة قبل أن ٌتم نشر القٌمة الجدٌدة إلى بقٌة المسارات وفً المثال السابق ٌتم التأكد من MemoryBarrierوٌجب علٌك استدعاء الطرٌقة 

  قبل أن توضع فً المتغٌر الذي ستتم مشاركته عبر المساراتtempInstanceاكتمال وضع القٌمة فً المتغٌر 

 

The Monitor Type 

 مسابل التزامن ولكنها تكون غٌر ملابمة فً العدٌد من الحالات   طرٌقة سهلة لاستخدام طرٌقة تتعامل معSyncLock توفر كتلة 

 مرتبطة مع نفس SyncLock  وتجنب منعه من ذلك إذا كان مسار آخر ٌنفذ كتلة SyncLock فمثلا لا ٌمكن للمسار اختبار كود فً كتلة 

 التً ٌمكن استخدامها مباشرة للحصول على مرونة أكثر Monitor objects معرفة داخلٌا بواسطة SyncLock  وكتل Objectالغرض 

 التً Monitor type وحٌد وفً الحقٌقة جمٌع طرق Monitor objectولا ٌمكنك استخدام . وٌتم ذلك على حساب زٌادة التعقٌد فً الكود

 الذي ٌعمل كالمحدد الممرر لكتلة Object هً الطرٌقة الأهم وهً تأخذ محدد من النوع Enterسٌتم عرضها هً طرابق ساكنة وتعتبر 

SyncLock وتكون له نفس الشروط من كونه من نوع مرجعً ومشترك ولا ٌمكن أن ٌحمل القٌمة Nothing وإن لم تمتلك المسارات 

 وإن امتلك مسار آخر القفل ٌجب على المسار 1الأخرى قفلا على هذا الغرض فٌقوم المسار الحالً بطلب ذلك القفل وٌضبط قٌمة العداد إلى 

الطالب انتظار أن ٌقوم المسار الآخر بتحرٌر القفل حتى ٌصبح متوفرا وإن كان المسار الطالب ٌمتلك القفل أساسا ٌؤدي كل استدعاء للطرٌقة 

Monitor.Enterوتأخذ الطرٌقة .  إلى زٌادة قٌمة العدادMonitor.Exit غرض القفل  lock object كمحدد لها وتنقص قٌمة العداد وعندما 

تصل قٌمة العداد للصفر ٌتم تحرٌر القفل ممكنا بقٌة المسارات من الحصول علٌه وٌجب أن ٌتم الموازنة بٌن استدعاء الطرٌقة 

Monitor.Enter والطرٌقة Monitor.Exitأو لن ٌتم تحرٌر القفل أبدا  

 
' A non-Nothing module-level object variable 

Dim objLock As New Object() 

… 

Try 

   ' Attempt to enter the protected section; 

   ' wait if the lock is currently owned by another thread. 

   Monitor.Enter(objLock) 

   ' Do something here. 

   … 

Finally 

   ' Release the lock. 

   Monitor.Exit(objLock) 

End Try 

 



 استثناء ٌجب علٌك عندها وضع كامل الكود ضمن كتلة Exit و Enterإذا كان هناك احتمال فً أن تطلق العبارات الموجودة بٌن الطرٌقتان 

Try…End Try لأنه من الضروري أن تقوم بتحرٌر القفل دوما وإن طلب مسار طرٌقة على مسار آخر تنتظر داخل الطرٌقة 

Monitor.Enter سوف ٌستقبل ذلك المسار استثناء ThreadInterruptedException الذي ٌعتبر سببا إضافٌا لاستخدام كتلة 

Try…End Try والطرٌقتان Enter و Exit الخاصتٌن بـ Monitor Object ٌسمحان لك باستبدال كتلة  SyncLock ولكنهما لا ٌقدمان لك 

 ولكنها تخرج Enter وهً مشابهة للطرٌقة TryEnter عندما تطبق الطرٌقة Monitorأٌة فوابد إضافٌة وسوف ترى المرونة الزابدة للفبة 

 مٌللً 10 خلال Monitor إذا كان لا ٌمكن الحصول على القفل خلال فترة زمنٌة محددة فمثلا ٌمكنك محاولة الحصول على Falseوتعٌد 

 SyncLock ثانٌة ثم التخلً عن ذلك دون أن توقف المسار الحالً مدة غٌر محددة وٌقوم الكود التالً بإعادة كتابة المثال السابق المعتمد على 

  وٌظهر لك المحاولات الفاشلة للحصول على القفلMonitor objectمستخدما 

 
Try 

   Do Until Monitor.TryEnter(consoleLock, 10) 

      Debug.WriteLine("Thread " + Thread.CurrentThread.Name + _ 

         " failed to acquire the lock") 

   Loop 

   ' Split the output line in pieces. 

   Console.Write(" ") 

   Console.Write(Thread.CurrentThread.Name) 

Finally 

   ' Release the lock. 

   Monitor.Exit(consoleLock) 

End Try 

 

The Mutex Type 

 ٌمكن امتلاكه من قبل مسار واحد Windows kernel object هو Mutex مبدأ آخر للتزامن حٌث أن الـ Mutex ٌوفر النوع 

 باستخدام Mutexوٌطلب المسار ملكٌة الـ .  عندما لا ٌمتلكه أي مسارa signaled stateفقط فً الوقت نفسه وٌكون فً حالة إشارة 

 والتً لا تعود إلا بعد أن ٌتم تحقٌق الملكٌة وٌتم تحرٌرها باستخدام الطرٌقة الساكنة Mutex.WaitOneالطرٌقة الساكنة 

Mutex.ReleaseMutex والمسار الذي ٌطلب ملكٌة Mutex object المملوك من قبله سلفا لا ٌمنع نفسه من الحصول على الملكٌة فٌجب 

 Mutex بعدد مساوي من المرات وهذا مثال عن كٌفٌة تعرٌف قسم متزامن باستخدام ReleaseMutexعلٌك فً هذه الحالة استدعاء 

object 

 
' This Mutex object must be accessible to all threads. 

Dim m As New Mutex() 

 

Sub WaitOneExample() 

   m.WaitOne() 

   ' Enter the synchronized section. 

   … 

   ' Exit the synchronized section. 

   m.ReleaseMutex() 

End Sub 

 
 وإن قمت Finally فً قسم ReleaseMutex لحماٌة كودك من الأخطاء ووضع استدعاء Tryوفً التطبٌقات الحقٌقٌة علٌك استخدام كتلة 

 كزمن انتهاء فستعٌد التحكم للمسار عندما ٌتم تحقٌق الملكٌة بنجاح أو عندما ٌنتهً الوقت المحدد WaitOneبتمرٌر محدد اختٌاري للطرٌقة 

  تعنً انتهاء الوقتFalse تعنً تحقٌق الملكٌة و Trueوٌمكن معرفة الفرق بٌن النتٌجتٌن باختبار القٌمة المعادة حٌث أن 

 
' Attempt to enter the synchronized section, but give up after 0.1 seconds. 

If m.WaitOne(100, False) Then 

 

   ' Enter the synchronized section. 

   … 

   ' Exit the synchronized section, and release the mutex. 

   m.ReleaseMutex() 



End If 

 
 بدون تقدٌم أٌة خصابص إضافٌة وٌمكنك رؤٌة Monitor.TryEnter آلٌة مكافبة للطرٌقة Mutexعند استخدام هذه الطرٌقة ٌوفر النوع 

 تأخذ مصفوفة WaitAny الخاصتٌن به والطرٌقة WaitAll  و WaitAny عندما ترى الطرٌقتٌن الساكنتٌن Mutexالمرونة الإضافٌة للنوع 

 فً حالة Mutex من تلك القابمة وفً هذه الحالة ٌصبح الـ Mutex objects وتعود عندما تحقق ملكٌة واحد من Mutex objectsمن 

 التً أصبحت فً Mutex objectsإشارة أو عندما ٌنتهً الوقت المحدد بالمحدد الاختٌاري والقٌمة المعادة تكون عبارة عن مصفوفة من 

 عندما ٌكون لدٌنا عدد محدود من Mutex objectsوتستخدم مصفوفة من .  عندما ٌنتهً الوقت المحدد258حالة إشارة أو قٌمة خاصة هً 

 الذي فً حالة Mutex objectsالموارد ونرٌد أن نربط كل واحد منها بمسار حالما ٌصبح ذلك المصدر متوفرا وفً هذه الحالة ٌصبح الـ 

 لمنع المسار الحالً حتى ٌصبح واحدا من الـ Mutex.WaitAnyإشارة ٌعنً أن المصدر الموافق متوفر عندبذ ٌمكنك استخدام الطرٌقة 

Mutex objects فً حالة إشارة و النوع Mutex ٌرث الطرٌقة WaitAny من WaitHandle الخاصة بفبته الأب وهذا هٌكل لتطبٌق 

 ٌستخدم هذه التقنٌة

 

' An array of three Mutex objects 

Dim mutexes() As Mutex = {New Mutex(), New Mutex(), New Mutex()} 

 

Sub WaitAnyExample() 

   ' Wait until a resource becomes available. 

   ' (Returns the index of the available resource.) 

   Dim mutexNdx As Integer = Mutex.WaitAny(mutexes) 

   ' Enter the synchronized section. 

   ' (This code should use only the resource corresponding to mutexNdx.) 

   … 

   ' Exit the synchronized section, and release the resource. 

   mutexes(mutexNdx).ReleaseMutex() 

End Sub 

 
 وتعٌد التحكم Mutex objects الخاصة بالفبة الأب حٌث تأخذ مصفوفة من WaitHandle أٌضا موروثة من WaitAllوالطرٌقة الساكنة 

للتطبٌق فقط عندما ٌصبح جمٌعهم فً حالة إشارة وهً مفٌدة بشكل خاص عندما لا ٌمكنك المتابعة إلا عندما تكون جمٌع المسارات الباقٌة قد 

 أنهت عملها

' Wait until all resources have been released. 

Mutex.WaitAll(mutexes) 

 
 Single هً أنه لا ٌمكن استدعاؤها من المسار الربٌسً فً تطبٌق مسار الغرفة الوحٌدة WaitAllوهناك مشكلة صغٌرة متعلقة بالطرٌقة 

Thread Apartment (STA) application مثل تطبٌق الكونسول Console application أو تطبٌق نماذج وٌندوز Windows 

Forms application ففً المسار الربٌسً لتطبٌق STA ٌجب علٌك التوقف حتى ٌتم تحرٌر مجموعة من الـ Mutex عندها ٌجب علٌك 

 على ذلك المسار لإٌقاف المسار الربٌسً حتى تعود الطرٌقة Thread.Join من مسار منفصل ثم استخدام الطرٌقة WaitAllاستخدام 

WaitAll من الفرٌموورك ٌوجد الطرٌقة الساكنة الجدٌدة 2 والنسخة 2005 وفً فٌجول باٌزٌك SignalAndWait تمكنك من وضع 

Mutex object فً حالة إشارة وانتظار Mutex objectآخر  

 
' Signal the first mutex and wait for the second mutex to become signaled. 

Mutex.SignalAndWait(mutexes(0), mutexes(1)) 

 
 أن ٌرتبط باسم الأمر الذي ٌعد من أهم المزاٌا لهذه Mutex objectsوخلافا لجمٌع أغراض التزامن التً تم ذكرها حتى الآن ٌمكن لـ 

  لها كما ٌلInstanceً التً تمتلك نفس الاسم ٌمكن مشاركتها عبر العملٌات وٌمكنك إنشاء تواجد Mutex objectsالأغراض فأغراض 

 

Dim m As New Mutex(False, "mutexname") 

 



 جدٌد بحٌث تمكنك هذه الآلٌة من Mutex objectوإن كان الاسم موجودا سابقا فً النظام ٌحصل المستدعً على مرجع له وإلا سٌتم إنشاء 

 عبر عدة تطبٌقات مختلفة وبهذا تتمكن هذه التطبٌقات من مزامنة عملٌات الوصول للمصادر المختلفة وقد تم Mutex objectsمشاركة 

 Mutex ٌمكنك من اختبار إذا كان قد تم منح المسار المستدعً ملكٌة الـ 2005 وفٌجول باٌزٌك 2إضافة بانً جدٌد فً الفرٌموورك 

 
Dim ownership As Boolean 

Dim m As New Mutex(True, "mutexname", ownership) 

If ownership Then 

   ' This thread owns the mutex. 

   … 

End If 

 
 هو تحدٌد فٌما إذا كان التطبٌق العامل هو الأول أو الوحٌد الذي تم تحمٌله وإن لم تكن هذه named mutexesمن الاستخدامات الشابعة لـ 

 الحالة ٌمكن للتطبٌق الخروج مباشرة أو الانتظار حتى تنتهً النسخة الأخرى من مهامها كما فً المثال

 
Sub Main() 

   Dim ownership As Boolean 

   Dim m As New Mutex(True, "DemoMutex", ownership) 

   If ownership Then 

      Console.WriteLine("This app got the ownership of Mutex named DemoMutex") 

      Console.WriteLine("Press ENTER to run another instance of this app") 

      Console.ReadLine() 

      Process.Start(Assembly.GetExecutingAssembly().GetName().CodeBase) 

   Else 

      Console.WriteLine("This app is waiting to get ownership of Mutex named 

DemoMutex") 

      m.WaitOne() 

   End If 

   ' Perform the task here. 

   … 

   Console.WriteLine("Press ENTER to release ownership of the mutex") 

   Console.ReadLine() 

   m.ReleaseMutex() 

End Sub 

 
 named على مستوى النظام Mutex وتقدم طرٌقة أخرى لفتح 2 جدٌدة أٌضا فً الفرٌموورك OpenExistingوالطرٌقة الساكنة 

system-wide Mutex object وبعكس بانً الـ Mutex تمكنك هذه الطرٌقة من تحدٌد درجة التحكم التً ترٌدها على الـ Mutex 

 

Try 

   ' Request a mutex with the right to wait for it and to release it. 

   Dim rights As MutexRights = MutexRights.Synchronize Or MutexRights.Modify 

   Dim m As Mutex = Mutex.OpenExisting("mutexname", rights) 

   ' Use the mutex here. 

   … 

Catch ex As WaitHandleCannotBeOpenedException 

   ' The specified object doesn't exist. 

Catch ex As UnauthorizedAccessException 

   ' The specified object exists, but current user doesn't have the 

   ' necessary access rights. 

Catch ex As IOException 

   ' A Win32 error has occurred. 

End Try 

 
 وهً إمكانٌة الوصول لقوابم التحكم بالوصول Mutex تظهر المٌزة الجدٌدة الأهم فً النوع 2 والفرٌموورك 2005وفً فٌجول باٌزٌك 

access control lists (ACLs) فً النموذج عبر الغرض System.Security.AccessControl.MutexSecurity object حٌث 



 MutexSecurity للحصول على غرض GetAccessControl جدٌد مستخدما الطرٌقة Mutex عندما تنشا غرض ACLٌمكنك تحدٌد 

 SetAccessControl جدٌد باستخدام الطرٌقة ACL محدد وتطبٌق Mutexالمرتبط بـ 

 

Dim ownership As Boolean 

Dim m As New Mutex(True, "mutexname", ownership) 

If Not ownership Then 

   ' Determine who is the owner of the mutex. 

   Dim mutexSec As MutexSecurity = m.GetAccessControl() 

   Dim account As NTAccount = DirectCast(mutexSec.GetOwner( _ 

   GetType(NTAccount)), NTAccount) 

   Console.WriteLine("Mutex is owned by {0}", account) 

End If 

 

The Semaphore Type 

 Win32 semaphoreالذي ٌرتكز على  Semaphore type  و فٌجول باٌزٌك دوت نٌت نوعا جدٌدا وهو2 تقدم الفرٌموورك 

object  وخلافا لبقٌة أغراض المسارات الموجودة فً المكتبة mscorlib فهذا النوع تم تعرٌفه فً المكتبة system.dll وهو ٌستخدم عندما 

من المسارات التً ٌمكن تنفٌذها فً جزء معٌن من الكود أو للوصول إلى مصدر معٌن وٌمتلك عددا ابتدابٌا  (Nعدد )ترٌد تحدٌد حد أقصى 

 وعددا أقصى وٌجب علٌك تمرٌر هذه القٌم لبانٌه

 
' A semaphore that has an initial count of 1 and a maximum count of 2. 

Dim sem As New Semaphore(1, 2) 

 
 وإن كان العدد الحالً أكبر من الصفر ٌتم إنقاصه وتعود الطرٌقة WaitOne باستدعاء الطرٌقة semaphoreٌحاول المسار أخذ ملكٌة الـ  

 semaphore أو إنقضاء الوقت المحدد بالمحدد الاختٌاري وٌحرر المسار semaphoreمباشرة وإلا تنتظر حتى ٌحرر مسار آخر 

  أو بقٌمة محددة وٌعٌد قٌمة العدد السابق1 مما ٌزٌد العدد بمقدار Releaseباستدعاء الطرٌقة 

 

Dim sem As New Semaphore(2, 2) 

' Next statement brings count from 2 to 1. 

sem.WaitOne() 

… 

' Next statement brings count from 1 to 2. 

sem.Release() 

' Next statement attempts to bring count from 2 to 3, but 

' throws a SemaphoreFullException. 

sem.Release() 

  كما ٌلSemaphoreًوبشكل أساسً ستستخدم الغرض 

 
' Initial count is initially equal to max count. 

Dim sem2 As New Semaphore(2, 2) 

 

Sub Semaphore_Example() 

   ' Wait until a resource becomes available. 

   sem2.WaitOne() 

   ' Enter the synchronized section. 

   … 

   ' Exit the synchronized section, and release the resource. 

   sem2.Release() 

End Sub 

 



 ٌمكن mutexes قد تم تحرٌره حتى لو حدث استثناء ما وتماما كالـ semaphore للتأكد من أن الـ Try…Finallyتذكر دوما استخدام الكتلة 

 موجود سابقا ٌتم تجاهل العدد والحد semaphores امتلاك اسم ومشاركته عبر العملٌات وعندما تحاول إنشاء غرض semaphoresللـ 

 الأقصى

Dim ownership As Boolean 

Dim sem3 As New Semaphore(2, 2, "semaphoreName", ownership) 

If ownership Then 

   ' Current thread has the ownership of the semaphore. 

   … 

End If 

 
 والتعدٌل GetAccessControl التً ٌمكن تمرٌرها للبانً حٌث تتم القراءة بواسطة الطرٌقة ACLs أٌضا الـ Semaphoreوٌدعم الغرض 

 تتم وراثتهما من الفبة الأساسٌة Semaphore والنوع Mutex أن تلاحظ أن النوع ي ومن الضرورSetAccessControlبواسطة الطرٌقة 

WaitHandle لذا ٌمكن تمرٌرهما كمحددات للطرابق الساكنة WaitAny   و WaitAll و  SignalAndWait للنوعWaitHandle مما 

 ٌمكنك من مزامنة المصادر بسهولة والتً تكون محمٌة بواسطة أٌا من هذه الأغراض كما فً الكود

 
' Wait until two mutexes, two semaphores, and one event object become signaled. 

Dim waitHandles() As WaitHandle = {mutex1, mutex2, sem1, sem2, event1} 

WaitHandle.WaitAll(waitHandles) 

 

The ReaderWriterLock Type 

 العدٌد من المصادر فً العالم الحقٌقً ٌمكن إما القراءة منها أو الكتابة إلٌها وهً تدعم فً الغالب إما مجموعة قراءات متعددة أو 

عملٌة كتابة وحٌدة ٌتم تنفٌذها فً لحظة معٌنة فمثلا ٌمكن لعدة عملاء القراءة من ملف بٌانات أو جدول فً قاعدة بٌانات ولكن إن تمت الكتابة 

للملف أو الجدول فلا ٌمكن حدوث أي عملٌات قراءة أو كتابة على ذلك المصدر حٌث ٌمكنك تعرٌف قفلا لكتابة واحدة أو عدة قراءات بدلالة 

 واستخدام هذا الغرض ٌعتبر رؤٌة إلى الأمام فكل المسارات التً ترٌد استخدام مصدر معٌن ٌجب علٌها ReaderWriterLockالغرض 

 وقبل محاولة القٌام بأي عملٌة على ذلك المصدر ٌجب على المسار استدعاء إما الطرٌقة ReaderWriterLockاستخدام نفس الغرض 

AcquireReaderLock أو الطرٌقة AcquireWriterLock وذلك اعتمادا على العملٌة التً ٌتم تنفٌذها وهذه الطرابق تقوم بإٌقاف المسار 

 أو الطرٌقة ReleaseReaderLockالحالً حتى ٌتم الحصول على ذلك المصدر وأخٌرا على المسار استدعاء الطرٌقة 

ReleaseWriterLock مسارات تقوم بعملٌة قراءة أو 10 عندما تنهً عملٌة القراءة أو الكتابة على ذلك المصدر والمثال التالً ٌقوم بإنشاء 

 كتابة على مصدر مشترك

 
Dim rwl As New ReaderWriterLock() 

Dim rnd As New Random() 

 

Sub TestReaderWriterLock() 

   For i As Integer = 0 To 9 

      Dim t As New Thread(AddressOf ReaderWriterLock_Task) 

      t.Start(i) 

   Next 

   … 

End Sub 

 

Sub ReaderWriterLock_Task(ByVal obj As Object) 

   Dim n As Integer = CInt(obj) 

   ' Perform 10 read or write operations. (Reads are more frequent.) 

   For i As Integer = 1 To 10 

      If rnd.NextDouble < 0.8 Then 

         ' Attempt a read operation. 

         rwl.AcquireReaderLock(Timeout.Infinite) 

         Console.WriteLine("Thread #{0} is reading", n) 

         Thread.Sleep(300) 

         Console.WriteLine("Thread #{0} completed the read operation", n) 

         rwl.ReleaseReaderLock() 

      Else 



 

         ' Attempt a write operation. 

         rwl.AcquireWriterLock(Timeout.Infinite) 

         Console.WriteLine("Thread #{0} is writing", n) 

         Thread.Sleep(300) 

         Console.WriteLine("Thread #{0} completed the write operation", n) 

         rwl.ReleaseWriterLock() 

      End If 

   Next 

End Sub 

 
وعندما تشغل هذا الكود سترى أن عدة مسارات ٌمكنها القراءة بنفس الوقت والمسار الذي ٌقوم بالكتابة ٌوقف جمٌع المسارات الأخرى وٌمكن 

 وذلك بقٌمة من النوع timeout أخذ محدد عبارة عن زمن انتهاء AcquireWriterLock و AcquireReaderLockللطرٌقتان 

TimeSpan أو بعدد من المٌللً ثانٌة وٌمكنك اختبار فٌما إذا تم الحصول على القفل بنجاح باستخدام الخصابص IsReaderLockHeld و 

IsWriterLockHeld القابلة للقراءة فقط إذا مررت قٌمة غٌر Timeout.Infinite 

 
' Attempt to acquire a reader lock for no longer than 1 second. 

rwl.AcquireWriterLock(1000) 

If rwl.IsWriterLockHeld Then 

   ' The thread has a writer lock on the resource. 

   … 

End If 

 
 والعودة ثانٌة لوضع القراءة UpgradeToWriterLockوالمسار الذي ٌمتلك قفل القراءة ٌمكنه الترقٌة إلى قفل للكتابة باستدعاء الطرٌقة 

 هً أنها أغراض خفٌفة ReaderWriterLock والشا الرابع بخصوص الأغراض Downgrade-FromWriterLockباستخدام الطرٌقة 

 و AcquireReaderLockبحٌث ٌمكن استخدامها عددا كبٌرا من المرات دون أن تؤثر على الأداء بشكل ملحوظ وبما أن الطرابق 

AcquireWriterLock تأخذان وقت انتهاء فالتطبٌق المصمم بشكل جٌد ٌجب أن لا ٌعانً من أقفال مٌتة ومع ذلك ٌمكن حصول حالة قفل 

 مٌت عندما ٌكون مساران ٌنتظران مصدرا محجوزا من قبل مسار لا ٌقوم بتحرٌره حتى انتهاء العملٌة الجارٌة

 

The Interlocked Type 

 بطرٌقة للقٌام بعملٌات دقٌقة لزٌادة أو إنقاص قٌمة متغٌر مشترك وهذه الفبة تعرض فقط طرابق Interlocked ٌزودنا النوع 

 انظر إلى الكود التالً (Objectلا نحتسب هنا ما تمت وراثته من )ساكنة 

 
' Increment and Decrement methods work with 32-bit and 64-bit integers. 

Dim lockCounter As Integer 

… 

' Increment the counter and execute some code if its previous value was zero. 

If Interlocked.Increment(lockCounter) = 1 Then 

   … 

End If 

' Decrement the shared counter. 

Interlocked.Decrement(lockCounter) 

 
  بت بقٌمة محددة64 أو 32 من عٌار Integer وهً تمكنك من زٌادة أعداد حقٌقٌة 2 جدٌدة فً الفرٌموورك ADDوالطرٌقة 

 
If Interlocked.Add(lockCounter, 2) <= 10 Then…  

 
 التً تمكنك من تحدٌد قٌمة من اختٌارك إلى متغٌرات من النوع Exchange طرٌقتان ساكنتان أخرٌان الطرٌقة Interlockedوتوفر الفبة 

Integer أو  Long أو  Single أو  Double أو  IntPtr أو  Object وتعٌد القٌمة السابقة وبما أن لها نسخة محملة زابدا تأخذ محددا من

  كما فً المثالString لهذا ٌمكنك أن تجعلها تعمل لأي نوع مرجعً كالنوع Objectالنوع 

 



Dim s1 As String = "123" 

Dim s2 As String = Interlocked.Exchange(s1, "abc") 

Console.WriteLine("s1={0}, s2={1}", s1, s2) 

 

  تعمل بأسلوب مشابهة ولكنها تقوم بالتبدٌل فقط إذا كان موقع الذاكرة مساوي لقٌمة محددة ٌتم تمرٌرها لهاCompareExchangeوالطرٌقة 

 

The ManualResetEvent, AutoResetEvent, and EventWaitHandle Types 

 
 والفبة الأخٌرة EventWaitHandle و AutoResetEvent  و ManualResetEvent هذه الفبات الثلاثة تعمل بشكل متشابه 

 لم AutoResetEvent و ManualResetEvent على الرغم من أن 2هً الفبة الأب للفبتان الأولان وقد تمت إضافتها فً الفرٌموورك 

والنوعان .  التً تعطٌك مزٌدا من المرونة عند التعاملEventWaitHandleٌتم إهمالهما بعد وأثناء العمل ٌمكنك استبدالهما بالفبة الجدٌدة 

ManualResetEvent و  AutoResetEvent مفٌدان بشكل خاص عندما ترٌد إٌقاف مسار أو أكثر بشكل مؤقت حتى ٌخبرنا مسار آخر

 فً أسمابهما Eventمتوقف ولكن لا تنخدع بوجود  بأنه لا مانع من المتابعة وتستخدمهما لإٌقاظ مسار مثل إجراء معالجة الحدث فً مسار

وكابن من أحد هذٌن النوعٌن ٌمكن أن ٌكون فً حالة إشارة أو عدم . فلا ٌمكنك استخدام إجراءات معالجة الحدث التقلٌدٌة مع هذه الأغراض

إٌقاف حٌث ستمرر الحالة الابتدابٌة / وهذه القٌمة لا تملك أي معنى خاص بحٌث ٌمكنك اعتبارها كحالة تشغٌلSignale/UnSignaledإشارة 

 Reset أو ٌستخدم الطرٌقة Set باستخدام الطرٌقة Signaledللبانً وأي مسار ٌستطٌع الوصول لذلك الغرض ٌمكنه ضبط تلك الحالة إلى 

 أو Signaled للانتظار حتى تصبح فً حالة إشارة WaitOne وٌمكن للمسارات الأخرى استخدام الطرٌقة UnSignaledلإعادة الحالة إلى 

 حتى انتهاء فترة الانتظار

 
' Create an auto reset event object in nonsignaled state. 

Dim are As New AutoResetEvent(False) 

' Create a manual reset event object in signaled state. 

Dim mre As New ManualResetEvent(True) 

 
ٌصبح فً حالة عدم ) هو أن الأخٌر ٌعٌد ضبط نفسه آلٌا AutoResetEvent و ManualResetEventوالاختلاف الوحٌد بٌن الغرضان 

 فقط واحد AutoResetEvent وٌوقظ الغرض WaitOneوذلك مباشرة بعد أن ٌتم صد المسار عندما تبدأ الطرٌقة   (Unsignaledإشارة 

 ٌوقظ جمٌع المسارات المنتظرة وٌجب أن ٌتم ManualResetEventمن المسارات المنتظرة عندما ٌصبح فً حالة إشارة بٌنما الغرض 

 و ManualResetEventإعادة ضبطه ٌدوٌا إلى حالة عدم إشارة كما هو ظاهر من اسمه وكما ذكر سابقا ٌمكنك استبدال الغرضٌن 

AutoResetEvent بالغرض EventWaitHandleًكما ٌظهر بالكود التال  

 
' These statements are equivalent to the previous code example. 

Dim are As New EventWaitHandle(False, EventResetMode.AutoReset) 

Dim mre As New EventWaitHandle(True, EventResetMode.ManualReset) 

 
 مفٌدة خاصة فً حالات المنتج والمستهلك فربما ٌكون لدٌك إجراء وحٌد فً مسار ٌقوم بتقٌٌم بعض البٌانات أو Eventوتكون أغراض الـ 

 على غرض متزامن فٌتم إعادة تشغٌل مسار أو أكثر لمعالجة تلك Setبالقراءة من القرص أو منفذ تسلسلً أو غٌرها وٌستدعً الطرٌقة 

 إذا كان هناك مسار AutoReset مع الخٌار EventWaitHandle أو الغرض AutoResetEventالبٌانات وٌجب علٌك استخدام الغرض 

 مع EventWaitHandle أو الغرض ManualResetEventمستهلك وحٌد سٌقوم بمعالجة تلك البٌانات كما ٌجب علٌك استخدام الغرض 

 .  إذا كان ٌجب معالجة البٌانات باستخدام جمٌع المسارات المستهلكةManualResetالخٌار 

وٌبٌن المثال التالً كٌف ٌمكن أن ٌكون لدٌك عدة مسارات منتجة تقوم بعملٌة البحث عن ملف فً عدة مجلدات مختلفة فً نفس الوقت ولكن 

 لإٌقاظ المسار المستهلك عندما AutoResetEventٌوجد مسار مستهلك وحٌد ٌقوم بجمع النتابج من تلك المسارات وٌستخدم المثال الغرض 

 لإدارة عدد المسارات العاملة حتى ٌعلم المسار Interlocked وٌستخدم أٌضا الفبة List(Of String)ٌتم إضافة اسم ملف جدٌد للقابمة 

 الربٌسً أنه لم تعد توجد أي بٌانات أخرى لاستهلاكها

 
' The shared AutoResetEvent object 

Public are As New AutoResetEvent(False) 

' The list where matching filenames should be added 

Public fileList As New List(Of String)() 



' The number of running threads 

Public searchingThreads As Integer 

' An object used for locking purposes 

Public lockObj As New Object() 

 

Sub TestAutoResetEvent() 

   ' Search *.zip files in all the subdirectories of C. 

   For Each dirname As String In Directory.GetDirectories("C:\") 

      Interlocked.Increment(searchingThreads) 

      ' Create a new wrapper class, pointing to a subdirectory. 

      Dim sf As New FileFinder() 

      sf.StartPath = dirname 

      sf.SearchPattern = "*.zip" 

      ' Create and run a new thread for that subdirectory only. 

      Dim t As New Thread(AddressOf sf.StartSearch) 

      t.Start() 

   Next 

 

   ' Remember how many results we have so far. 

 

   Dim resCount As Integer = 0 

   Do While searchingThreads > 0 

      ' Wait until there are new results. 

      are.WaitOne() 

 

      SyncLock lockObj 

         ' Display all new results. 

         For i As Integer = resCount To fileList.Count - 1 

             Console.WriteLine(fileList(i)) 

         Next 

         ' Remember that you've displayed these filenames. 

         resCount = fileList.Count 

      End SyncLock 

   Loop 

   Console.WriteLine("") 

   Console.WriteLine("Found {0} files", resCount) 

End Sub 

 
 مختلف الذي ٌجب أن ٌكون قادرا على الوصول إلى متغٌرات عامة محددة فً الكود FileFinderوكل مسار إجرابً ٌعمل ضمن غرض 

 السابق

 
Class FileFinder 

   Public StartPath As String       ' The starting search path 

   Public SearchPattern As String   ' The search pattern 

 

   Sub StartSearch() 

      Search(Me.StartPath) 

      ' Decrease the number of running threads before exiting. 

      Interlocked.Decrement(searchingThreads) 

      ' Let the consumer know it should check the thread counter. 

      are.Set() 

   End Sub 

 

   ' This recursive procedure does the actual job. 

   Sub Search(ByVal path As String) 

      ' Get all the files that match the search pattern. 

      Dim files() As String = Directory.GetFiles(path, SearchPattern) 

      ' If there is at least one file, let the main thread know about it. 

      If files IsNot Nothing AndAlso files.Length > 0 Then 

         ' Ensure found files are added as an atomic operation. 

         SyncLock lockObj 

            ' Add all found files. 



            fileList.AddRange(files) 

            ' Let the consumer thread know about the new filenames. 

            are.Set() 

         End SyncLock 

      End If 

 

      ' Repeat the search on all subdirectories. 

      For Each dirname As String In Directory.GetDirectories(path) 

         Search(dirname) 

      Next 

   End Sub 

End Class 

 
 أو AutoResetEvent بدلا عن EventWaitHandle تستخدم 2005 من الفرٌموورك وفٌجول باٌزٌك 2والنسخة 

ManualResetEvent مما ٌعطٌك مٌزة هامة وهً إمكانٌة إنشاء غرض مسمى على مستوى النظام ٌمكن مشاركته مع العملٌات الأخرى 

 Mutex مشابهة لتلك الخاصة بالفبة EventWaitHandleوالصٌغة العامة لبانً 

 
Create a system-wide auto reset event that is initially in the signaled state. 

Dim ownership As Boolean 

Dim ewh As New EventWaitHandle(True, EventResetMode.AutoReset, "eventname", 

ownership) 

If ownership Then 

   ' The event object was created by the current thread. 

   … 

End If 

 
  لفتح غرض حدث موجودOpenExistingكما ٌمكنك استخدام الطرٌقة 

 
' This statement throws a WaitHandleCannotBeOpenedException if the specified 

' event doesn't exist, or an UnauthorizedAccessException if the current 

' user doesn't have the required permissions. 

ewh = EventWaitHandle.OpenExisting("eventname", _ 

               EventWaitHandleRights.FullControl) 

 

 
 SetAccessControl باستخدام الطرابق ACLs هً دعم 2 فً الفرٌموورك event objectsوالمٌزة الأخرى الهامة فً أغراض الأحداث 

 حٌث ٌمكنك استخدامه بنفس طرٌقة استخدام EventWaitHandleSecurity والتً تأخذ وتعٌد كابن من النوع GetAccessControlو 

  ACLs وأغراض الدوت نٌت الأخرى التً تدعم Mutexمثٌلاتها فً الغرض 

 

 

 
  



 كيفيت حنفيز عوليت في هساس آخش وإظهاس النخيجت في الخحكواث على النوىرج
 

سألنً أحد الإخوة عن مشكلة واجهته عند تنفٌذ عملٌة معٌنة على مسار آخر ومحاولته إظهار النتٌجة على النموذج فإذا افترضنا أنه 

 غٌر المسار الربٌسً الذي تنفذ علٌه عملٌات البرنامج وأن ذلك الإجراء Threadلدٌنا إجراء بسٌطا ٌنفذ عملٌة ما ونفذناه على مسار آخر 

  لصندوق نصوص على النموذج فعند تنفٌذ الكود ستحصل على رسالة خطأTextٌحتوي على كود ٌقوم بضبط قٌمة الخاصٌة 

Cross-thread operation not valid: Control 'TextBox1' accessed from a thread other than the thread it was 

created on. 

وإذا أردت تولٌد رسالة الخطأ السابق بنفسك أنشا مشروعا جدٌدا ضع علٌه صندوق نصوص وزر واجعل كود النموذج مطابقا  لما ٌلً ثم 

 قم بتشغٌل البرنامج وستحصل على رسالة الخطأ السابقة

Imports System.Threading 

 

Public Class Form1 

 

    Private Sub Button1_Click() Handles Button1.Click 

 

        Dim th As New Thread(AddressOf DoLongOperation) 

        th.Start() 

 

    End Sub 

 

    Private Sub DoLongOperation() 

        Me.TextBox1.Text = "Something" 

    End Sub 

 

End Class 

 

 تقوم بتنفٌذ العملٌة على المسار الثانً وتعٌد النتٌجة للنموذج من خلال Classالحل الذي أقوم باستخدامه عادة لحل هكذا مشكلة هو إنشاء فبة 

إطلاق حدث ٌعٌد القٌم الناتجة عن عملٌة المعالجة للنموذج فربما لا تكون هذه هً الطرٌقة الأفضل فً جمٌع الحالات ولكنها طرٌقتً على 

كل حال وسأقوم بشرحها ثم ٌمكننا النقاش وتجربة أٌة حلول أخرى لتجاوز هذه المشكلة وسأطرحها عبر تنفٌذ عداد بسٌط للوقت فربما 

ستستخدم أنت هذه الطرٌقة للبحث عن ملفات أو تنفٌذ عملٌات معالجة معقدة تستغرق وقتا طوٌلا ولكننً هنا اخترت مثالا ٌعٌد قٌمة وحٌدة 

 بحٌث ٌكون بسٌطا قدر الإمكان

 جدٌد للمشروع ٌتم عبره تنفٌذ العملٌة الطوٌلة التً نرٌد تنفٌذها على مسار آخر وسأقوم بتسمٌتها Classالآن سأقوم بإضافة فبة 

MyStopWatchفً الوقت الحالً وبما أننا سنتعامل مع المسارات سنحتاج للاستٌراد التالً قبل تعرٌف الفبة  

Imports System.Threading 

 سأستخدمها لاحقا لإطلاق الحدث الذي سٌعٌد ReturnValueEventArgs باسم MyStopWatchسأقوم بتعرٌف فبة فرعٌة داخل الفبة 

 بما أنها فبة خاصة بإعادة قٌم الحدث الذي سٌتم إطلاقه وسأعرف EventArgsالنتٌجة إلى النموذج وهذه ٌجب أن تكون موروثة من الفبة 

 ستكون للقراءة فقط بما أننا لن نحتاج لضبط قٌمتها إلا من خلال بانً الفبة تعٌد القٌمة وبانً للفبة ٌمرر ReturnVlaueفٌها خاصٌة وحٌدة 

  كما ٌلReturnValueEventArgsًله قٌمة نصٌة وحٌدة تمثل القٌمة المعادة وبهذا ٌكون كود الفبة 

Public Class ReturnValueEventArgs 

    Inherits EventArgs 

 

    Private _ReturnValue As String 

 

    Public ReadOnly Property ReturnVlaue() As String 

        Get 

            Return _ReturnValue 

        End Get 

    End Property 



 

    Public Sub New(ByVal RetVal As String) 

        _ReturnValue = RetVal 

    End Sub 

 

End Class 

 

 نقوم بتعرٌف الحدث الذي سنقوم بإطلاقه لٌعٌد ReturnValueEventArgs وبعد نهاٌة تعرٌف الفبة MyStopWtachضمن كود الفبة 

 وبهذا ReturnValueEventArgsالقٌمة إلى النموذج ومن أجل الالتزام بتنسٌق الأحداث كما نرى فً التحكمات والفبات قمنا بتعرٌف الفبة 

  كما ٌلMyStopWatchًٌكون تعرٌف الحدث فً قسم تعرٌف المتغٌرات العامة فً الفبة 

Public Event ReturnValue(ByVal sender As Object, ByVal e As ReturnValueEventArgs) 

  كما ٌلStopwatchً وهو من النوع MyTimer_ باسم MyStopWtachعرف متغٌرا عاما على مستوى الفبة 

Private _MyTimer As Stopwatch 

حٌث سنستخدمه كعداد للوقت من أجل الحصول على قٌمة لٌتم إعادتها ضمن إجراء المعالجة الذي سٌتم تنفٌذه على المسار الآخر بحٌث 

 سٌكون كود إجراء المعالجة الذي سٌنفذ على المسار الثانً كما ٌلً

Private Sub DoProcessing() 

    Do 

        Dim Ret = _MyTimer.Elapsed.Hours & ":" & _ 

                  _MyTimer.Elapsed.Minutes & ":" & _ 

                  _MyTimer.Elapsed.Seconds & ":" & _ 

                  _MyTimer.Elapsed.Milliseconds 

 

        RaiseEvent ReturnValue(Me, New ReturnValueEventArgs(Ret)) 

    Loop Until _MyTimer.IsRunning = False 

End Sub 

 

 التً تشٌر Me حٌث القٌمة ReturnValue لإطلاق الحدث RaiseEvent ثم استخدمنا الدالة  Retحٌث وضعنا قٌمة العداد فً متغٌر نصً 

 من الفبة Instance ٌكون المحدد الثانً للحدث عبارة عن كٌان ReturnValueللفبة الحالٌة كبارمتر أول للحدث 

ReturnValueEventArgs التً نمرر لبانٌها المتغٌر Ret الذي ٌشكل القٌمة المعادة من الخاصٌة ReturnVlaue العابدة للفبة 

ReturnValueEventArgsعندما سنستقبلها من النموذج  

  بحٌث ٌكون كوده على الشكل StartTimerوسٌكون لدٌنا إجراء لبدء تنفٌذ المؤقت على المسار الثانً باسم 

Public Sub StartTimer() 

    _MyTimer.Reset() 

    _MyTimer.Start() 

 

    Dim th As New Thread(AddressOf DoProcessing) 

    th.Start() 

End Sub 

 

 ومن أجل إٌقاف العداد سنحتاج لإجراء DoProcessing ٌقوم بتنفٌذ الإجراء thحٌث قمنا بتصفٌر العداد وبدبه ثم عرفنا مسارا جدٌدا 

StopTimerٌكون كوده على الشكل  

Public Sub StopTimer() 

    _MyTimer.Stop() 

End Sub 

 

وبهذا تكون قد اكتملت فبتنا التً ستقوم بعملٌة المعالجة على مسار ثانً وتعٌد قٌمة نصٌة سنقوم بعرضها فً صندوق نصوص لاحقا وٌكون 

 بذلك الكود الكامل لهذه الفبة



Imports System.Threading 

 

Public Class MyStopWtach 

 

    Public Class ReturnValueEventArgs 

        Inherits EventArgs 

 

        Private _ReturnValue As String 

 

        Public ReadOnly Property ReturnVlaue() As String 

            Get 

                Return _ReturnValue 

            End Get 

        End Property 

 

        Public Sub New(ByVal RetVal As String) 

            _ReturnValue = RetVal 

        End Sub 

 

    End Class 

 

    Public Event ReturnValue(ByVal sender As Object, _ 

                   ByVal e As ReturnValueEventArgs) 

 

    Private _MyTimer As New Stopwatch 

 

    Public Sub StartTimer() 

        _MyTimer.Reset() 

        _MyTimer.Start() 

 

        Dim th As New Thread(AddressOf DoProcessing) 

        th.Start() 

    End Sub 

 

    Private Sub DoProcessing() 

        Do 

            Dim Ret = _MyTimer.Elapsed.Hours & ":" & _ 

                      _MyTimer.Elapsed.Minutes & ":" & _ 

                      _MyTimer.Elapsed.Seconds & ":" & _ 

                      _MyTimer.Elapsed.Milliseconds 

 

            RaiseEvent ReturnValue(Me, New ReturnValueEventArgs(Ret)) 

        Loop Until _MyTimer.IsRunning = False 

    End Sub 

 

    Public Sub StopTimer() 

        _MyTimer.Stop() 

    End Sub 

 

 

End Class 

 

نعود للنموذج الخاص بالمشروع الذي نحتاج لوجود صندوق نصوص وزرٌن علٌه للقٌام بتجربة الفبة الجدٌدة حٌث سنقوم بتعرٌف متغٌر 

 التً ستمكننا من استقبال WithEvents وباستخدام العبارة MyStopWtach من نوع فبتنا MyTimerخاص على مستوى النموذج باسم 

 الأحداث التً ستطلقها فبتنا

Private WithEvents MyTimer As New MyStopWtach 

 وسٌكون كود الزرٌن من أجل بدء وإٌقاف المؤقت باستخدام فبتنا السابقة كما ٌلً

Private Sub Button1_Click() Handles Button1.Click 

    MyTimer.StartTimer() 



End Sub 

 

Private Sub Button2_Click() Handles Button2.Click 

    MyTimer.StopTimer() 

End Sub 

 

 واجعله بحٌث ٌكون الكود فٌه كالتالً ثم جرب تشغٌل البرنامج MyTimer العابد للمتغٌر ReturnVlaueالآن أنشا معالج للحدث 

 فستحصل على رسالة مشابهة للرسالة فً بداٌة المقال

Private Sub MyTimer_ReturnValue(ByVal sender As Object, _ 

      ByVal e As MyStopWtach.ReturnValueEventArgs) Handles MyTimer.ReturnValue 

 

    Me.TextBox1.Text = e.ReturnVlaue 

 

End Sub 

 

 حتى نستطٌع استخدام القٌم MyTimer_ReturnValue للإجراء Invokeولمعالجة هذه النقطة والتخلص من رسالة الخطأ سنحتاج لعمل 

  لصندوق النصوص والعملٌة ببساطة ستتم كما ٌلTextًالمعادة منه فً ضبط قٌم خصابص التحكمات على النموذج وفً حالتنا هنا الخاصٌة 

 MyTimer_ReturnValue ٌحمل نفس توقٌع الإجراء Delegateفً قسم المتغٌرات العامة فً النموذج سنقوم بتعرٌف إجراء مفوض 

 وبدون جسم للإجراء كما ٌلً

Private Delegate Sub MyTimer_ReturnValueDelegate(ByVal sender As Object, _ 

                             ByVal e As MyStopWtach.ReturnValueEventArgs) 

 

 وٌكون  الكود الذي سٌنفذ المهمة بصورة صحٌحة كما ٌلً

Private Sub MyTimer_ReturnValue(ByVal sender As Object, _ 

        ByVal e As MyStopWtach.ReturnValueEventArgs) Handles MyTimer.ReturnValue 

 

    If Me.TextBox1.InvokeRequired = True Then 

        Dim d As New MyTimer_ReturnValueDelegate(AddressOf MyTimer_ReturnValue) 

        Me.Invoke(d, New Object() {sender, e}) 

    Else 

        Me.TextBox1.Text = e.ReturnVlaue 

    End If 

End Sub 

 

 Text نقوم بضبط قٌمة الخاصٌة False لصندوق النصوص فإن كان InvokeRequiredحٌث فحصنا قٌمة الخاصٌة القابلة للقراءة فقط 

 عندها لن نستطٌع ضبط القٌمة مباشرة كً لا نحصل على الخطأ Trueباستخدام القٌمة المعادة من الحدث مباشرة بدون أي مشاكل وإن كانت 

 ونمرر له عنوان MyTimer_ReturnValueDelegate من نوع الإجراء المفوض dالوارد فً بداٌة المقال عندها سنعرف متغٌر 

 العابدة للنموذج لتنفٌذ نسخة آمنة من الحدث Invoke ثم استخدمنا الطرٌقة MyTimer_ReturnValueالإجراء 

MyTimer.ReturnValue تمكننا من ضبط القٌم المعادة إلى التحكمات وذلك بتمرٌر المتغٌر d كمحدد أول للخاصٌة Invoke وٌكون 

  كعناصر لها MyTimer_ReturnValue ٌتم تمرٌر محددات الإجراء Object هو مصفوفة من النوع Invokeالمحدد الثانً للخاصٌة 

 وفٌما ٌلً الكود الكامل للنموذج

Public Class Form1 

 

    Private Delegate Sub MyTimer_ReturnValueDelegate(ByVal sender As Object, _ 

                               ByVal e As MyStopWtach.ReturnValueEventArgs) 

 

    Private WithEvents MyTimer As New MyStopWtach 

 

    Private Sub Button1_Click() Handles Button1.Click 

        MyTimer.StartTimer() 



    End Sub 

 

    Private Sub Button2_Click() Handles Button2.Click 

        MyTimer.StopTimer() 

    End Sub 

 

    Private Sub MyTimer_ReturnValue(ByVal sender As Object, _ 

             ByVal e As MyStopWtach.ReturnValueEventArgs) Handles _ 

              MyTimer.ReturnValue 

 

        If Me.TextBox1.InvokeRequired = True Then 

            Dim d As New MyTimer_ReturnValueDelegate( _ 

             AddressOf MyTimer_ReturnValue) 

 

            Me.Invoke(d, New Object() {sender, e}) 

        Else 

            Me.TextBox1.Text = e.ReturnVlaue 

        End If 

    End Sub 

 

End Class 

 

 


